
**APPENDIX TABLE 5 : Pvalues**

local mht = 1
local mht_pair = 1
local reps = 3000 //set reps for bootstrapping SE
log using "$out\OA-TableA5-Pvals.txt", text replace
log off

**  Initial data steps
	use "$data\merge_bl_fu", clear 
	gen A_EDT_elig=(A_EDT==1 & d_lleligible==1)
	gen A_EDT_no_elig=(A_EDT==1 & d_lleligible==0)
	gen nobt_ne_nolw = EDT_treat==0 & d_lleligible==0 & L_ttreat==0

	gen treat=.
	replace treat=1 if A_EDT==1 
	replace treat=2 if B_LL
	replace treat=3 if C_EDT_LL
	replace treat=0 if EDT_treat==0 & L_ttreat==0
	recode d_lleligible 0=2

	gen treatlong =.
	replace treatlong=1 if A_EDT==1 & d_lleligible == 1 & B_LL == 0 
	replace treatlong=2 if A_EDT==1 & d_lleligible == 2
	replace treatlong=3 if A_EDT==0 & d_lleligible == 1 & B_LL == 1 
	replace treatlong=4 if C_EDT_LL == 1
	replace treatlong=5 if A_EDT==0 & d_lleligible == 2 
	replace treatlong=0 if A_EDT==0 & d_lleligible == 1 & B_LL == 0 & treatlong ==.

**  Household expenditures and assets
	foreach var in log_monthly_expenditure log_savings _index_house log_livestock_value {
		log on
		reg `var'_fu `var'_bl A_EDT_elig A_EDT_no_elig  B_LL C_EDT_LL nobt_ne_nolw female bl_bus_owner fu_dummy*, cl(co_code1)
		log off
		gen xbhat = _b[_cons] +  _b[`var'_bl]*`var'_bl + _b[female]*female + ///
			_b[fu_dummy1]*fu_dummy1 + _b[fu_dummy2]*fu_dummy2 + _b[fu_dummy3]*fu_dummy3 + ///
			_b[fu_dummy5]*fu_dummy5 + _b[fu_dummy6]*fu_dummy6 + _b[bl_bus_owner ]*bl_bus_owner
		gen `var'_r = `var'_fu - xbhat
		drop xbhat
	}

	if `mht' > 0 {
		log on
		mhtexp log_monthly_expenditure_r log_savings_r  _index_house_r log_livestock_value_r, treatment(treatlong) bootstrap(`reps')
		log off
	}
	if `mht_pair' > 0 {
		log on
		mata: OnlyHyp = 1,1,4 \ 1,1,5 \ 1,1,6 \ 2,1,4 \ 2,1,5 \ 2,1,6 \ 3,1,4 \ 3,1,5 \ 3,1,6 \ 4,1,4 \ 4,1,5 \ 4,1,6 
		mhtexp log_monthly_expenditure_r log_savings_r  _index_house_r log_livestock_value_r, treatment(treat) subgroup(d_lleligible) combo("pairwise") only(OnlyHyp) bootstrap(`reps')
		log off 
	}


**  CO Cohesion
	foreach var in more_help more_collective {
		log on
		reg `var'_fu `var'_bl A_EDT_elig A_EDT_no_elig  B_LL C_EDT_LL nobt_ne_nolw female bl_bus_owner fu_dummy*, cl(co_code1)
		log off
		gen xbhat = _b[_cons] +  _b[`var'_bl]*`var'_bl +_b[female]*female + ///
			_b[fu_dummy1]*fu_dummy1 + _b[fu_dummy2]*fu_dummy2 + _b[fu_dummy3]*fu_dummy3 + ///
			_b[fu_dummy5]*fu_dummy5 + _b[fu_dummy6]*fu_dummy6 + _b[bl_bus_owner ]*bl_bus_owner
		gen `var'_r = `var'_fu - xbhat
		drop xbhat
	}

	foreach var of varlist d_give_loan08_fu d_take_loan08_fu {
		log on
		reg `var' A_EDT_elig A_EDT_no_elig  B_LL C_EDT_LL nobt_ne_nolw female bl_bus_owner fu_dummy*, cl(co_code1)
		log off
		gen xbhat = _b[_cons] +_b[female]*female + _b[fu_dummy1]*fu_dummy1 + ///
			_b[fu_dummy2]*fu_dummy2 + _b[fu_dummy3]*fu_dummy3 + _b[fu_dummy5]*fu_dummy5 + ///
			_b[fu_dummy6]*fu_dummy6 + _b[bl_bus_owner ]*bl_bus_owner
		gen `var'_r = `var' - xbhat
		drop xbhat
	}

	if `mht' > 0 {
		log on
		mhtexp more_help_r more_collective_r d_give_loan08_fu_r d_take_loan08_fu_r, treatment(treatlong) bootstrap(`reps')
		log off
	}
	if `mht_pair' > 0 {
		log on
		mata: OnlyHyp = 1,1,4 \ 1,1,5 \ 1,1,6 \ 2,1,4 \ 2,1,5 \ 2,1,6 \ 3,1,4 \ 3,1,5 \ 3,1,6 \ 4,1,4 \ 4,1,5 \ 4,1,6 
		mhtexp more_help_r more_collective_r d_give_loan08_fu_r d_take_loan08_fu_r, treatment(treat) subgroup(d_lleligible) combo("pairwise") only(OnlyHyp) bootstrap(`reps')
		log off
	}


**  General outlook on life
	foreach var in index_trust index_optimism _optimism8_f5{
		log on
		reg `var'_fu `var'_bl A_EDT_elig A_EDT_no_elig  B_LL C_EDT_LL nobt_ne_nolw female bl_bus_owner fu_dummy*, cl(co_code1)
		log off
		gen xbhat = _b[_cons] + _b[`var'_bl]*`var'_bl +  _b[female]*female + ///
			_b[fu_dummy1]*fu_dummy1 + _b[fu_dummy2]*fu_dummy2 + _b[fu_dummy3]*fu_dummy3 + ///
			_b[fu_dummy5]*fu_dummy5 + _b[fu_dummy6]*fu_dummy6 + _b[bl_bus_owner ]*bl_bus_owner
		gen `var'_r = `var'_fu - xbhat
		drop xbhat
	}

	if `mht' > 0 {
		log on
		mhtexp index_trust_r index_optimism_r _optimism8_f5_r, treatment(treatlong) bootstrap(`reps')
		log off
	}
	if `mht_pair' > 0 {
		log on
		mata: OnlyHyp = 1,1,4 \ 1,1,5 \ 1,1,6 \ 2,1,4 \ 2,1,5 \ 2,1,6 \ 3,1,4 \ 3,1,5 \ 3,1,6 
		mhtexp index_trust_r index_optimism_r _optimism8_f5_r, treatment(treat) subgroup(d_lleligible) combo("pairwise") only(OnlyHyp) bootstrap(`reps')
	}

log close
